Manipulación de polinomios e funcións racionais

Polinomios dunha variabel

Chámase forma canónica dun polinomio a:

$a_n x^n + a_{n-1} x^{n-1} + \ldots + a_1 x + a_0$

onde n é o grao do polinomio, a_n o primeiro coeficiente e a_0 o último. A definición dun polinomio farase mediante o operador de asignación

```
> p1:=-3*x + 7*x^2 - 3*x^3 + 7*x^4;
```

$$p1 := 7x^4 - 3x^3 + 7x^2 - 3x \quad (1.1.1)$$

Para avaliar $p_1(x_0=1)$:

```
> eval(p1,x=1);
```

$$8 \quad (1.1.2)$$

```
> type(p1, 'polynom'); # devolve true se p1 é un polinomio
```

$$\mathit{true} \quad (1.1.3)$$

```
> lcoeff(p1); # devolve o coeficiente do térmo de maior grado
```

$$7 \quad (1.1.4)$$

```
> degree(p1); #devolve o grao do polinomio
```

$$4 \quad (1.1.5)$$

```
> p2:=5*x^5 + 3*x^3 + x^2 -2*x + 1; # definición do polinomio p2
```

$$p2 := 5x^5 + 3x^3 + x^2 - 2x + 1 \quad (1.1.6)$$

```
> 2*p1 + 4*p2 +3;
```

$$20x^5 + 14x^4 + 6x^3 + 18x^2 - 14x + 7 \quad (1.1.7)$$

```
> p1*p2;
```

$$(7x^4 - 3x^3 + 7x^2 - 3x)(5x^5 + 3x^3 + x^2 - 2x + 1) \quad (1.1.8)$$

```
> expand(%); # calcula a expresión anterior
```

$$35x^9 - 15x^8 + 56x^7 - 17x^6 + 4x^5 + 11x^4 - 20x^3 + 13x^2 - 3x \quad (1.1.9)$$

```
> sort(%); # para obter o polinomio en forma canónica
```

$$35x^9 - 15x^8 + 56x^7 - 17x^6 + 4x^5 + 11x^4 - 20x^3 + 13x^2 - 3x \quad (1.1.10)$$

```
> coeff(p1,x^3); #devolve o coeficiente de x^3 do polinomio p1
```

$$-3 \quad (1.1.11)$$

Algúns comandos para a manipulación de polinomios que xa vimos son: **coeff( )** e **degree( )** (que só poden operar sobre polinomios en forma agrupada).

En MAPLE pódese calcula-lo cociente e o resto dunha división de polinomios coas funcións **quo** e **rem**:

rem(a, b, x) e **rem(a, b, x, 'q')** devolve o resto de dividir a entre b

quo(a, b, x) e **quo(a, b, x, 'r')** devolve o cociente de dividir a entre b

a e **b** son polinomios na variabel **x** e **'q'** e **'r'** son opcionais e representan os nomes das variabeis as que se asignará o resultado das respectivas operacións (resto para o comando quo e cociente para o comando rem).
O resto **r** e o cociente **q** cumpren: **a = b*q + r** onde **degree(r, x) < degree(b, x)**

```
> q1:=quo(p2, p1, x, 'r1'); # devolve o cociente e en r1 almacena o resto
```

$$q1 := \frac{5}{7}\,x + \frac{15}{49} \tag{1.1.12}$$

```
> r1;
```

$$-\frac{53}{49}\,x^3 + x^2 - \frac{53}{49}\,x + 1 \tag{1.1.13}$$

```
> testeq(p2=expand(q1*p1+r1)); #comprobación do resultado anterior
```

true **(1.1.14)**

```
> r2:=rem(p2, p1, x, 'q2'); # devolve o resto e almacena o cociente en q2
```

$$r2 := -\frac{53}{49}\,x^3 + x^2 - \frac{53}{49}\,x + 1 \tag{1.1.15}$$

```
> q2;
```

$$\frac{5}{7}\,x + \frac{15}{49} \tag{1.1.16}$$

```
> testeq(p2=expand(q2*p1+r2));
```

true **(1.1.17)**

A función **divide** devolve verdadeiro (*true*) cando a división entre dous polinomios é exacta (resto cero), i.e., *p1* divide a *p2*, e senón devolve falso (*false*).

```
> divide(p2, p1);
```

false **(1.1.18)**

Para calcula-lo máximo común divisor de dous polinomios utilízase a función **gcd** e para o mínimo común múltiplo a función **lcm**

gcd(a, b) **lcm(a, b,...)**

onde **a** e **b** son os polinomios.

```
> gcd(p2, p1);
```

$$x^2 + 1 \tag{1.1.19}$$

```
> lcm(p2,p1);
```

$$(5\,x^3 - 2\,x + 1)\,(7\,x^4 - 3\,x^3 + 7\,x^2 - 3\,x) \tag{1.1.20}$$

```
> expand(%); # calcula a expresión anterior
```

$$35\,x^7 - 15\,x^6 + 21\,x^5 - 2\,x^4 - 17\,x^3 + 13\,x^2 - 3\,x \tag{1.1.21}$$

```
> lcm(6, -8, 3, 4, 12); # lcm pódese aplicar a máis de dous factores
```

24 **(1.1.22)**

Tamén se pode calcular as raíces dun polinomio coa función **roots**. Só da as **raíces exactas** (enteiras, racionais, certas raíces irracionais, e raíces complexas con partes real/imaxinaria enteira, racional ou certas irracionais).

roots(a, x, K)

e devolve unha lista de pares da forma **[[r1, m1], ..., [rn, mn]]** onde **ri** e a raíz do polinomio **a** con multiplicidade **mi**, é dicir, **(x - ri)^{mi}** divide ao polinomio **a**.O argumento **K** indica o conxunto no que se buscan as raíces (por defecto, Z ou <; se se especifica **{sqrt(2)}**, búscanse nos irracionais que se poden poñer en termos de $\sqrt{2}$ (ídem se se pon **sqrt(3)**, etc); se se especifica **I**, busca nos números complexos con partes real e imaxinaria enteiras, racionais ou sqrt(2), etc.

```
> p3:=expand(p1*p2);
```

$$p3 := 35\,x^9 - 15\,x^8 + 56\,x^7 - 17\,x^6 + 4\,x^5 + 11\,x^4 - 20\,x^3 + 13\,x^2 - 3\,x \quad \textbf{(1.1.23)}$$

```
> roots(p3);
```

$$\left[[0, 1], \left[\frac{3}{7}, 1\right]\right] \quad \textbf{(1.1.24)}$$

Devolve só dúas raíces cando o polinomio é de grao 9. Esto débese a que só devolve as raíces racionais. Se queremos outras raíces hai que especifica-lo campo (distinto do campo dos racionais)

```
> roots(x^4-4, x); #ningunha raíz exacta racional
```

$$[\,] \quad \textbf{(1.1.25)}$$

```
> roots(x^4-4, sqrt(2)); # raices reais múltiplos de sqrt(2)
```

$$[[-\sqrt{2}, 1], [\sqrt{2}, 1]] \quad \textbf{(1.1.26)}$$

```
> roots(x^4-4, {sqrt(2), I}); # raices reais e múltiplos de
  sqrt(2)
```

$$[[-I\sqrt{2}, 1], [I\sqrt{2}, 1], [-\sqrt{2}, 1], [\sqrt{2}, 1]] \quad \textbf{(1.1.27)}$$

Pódese factorizar un polinomio (escribir un polinomio como producto de factores irreducibeis con coeficientes racionais) coa función **factor**:

factor(a)   **factor(a, K)**

a é o polinomio e K un argumento opcional igual que en roots.

```
> factor(p3);
```

$$x\,(7\,x-3)\,(5\,x^3-2\,x+1)\,(x^2+1)^2 \quad \textbf{(1.1.28)}$$

Polinomios de varias variabeis

MAPLE permite definir polinomios de varias variabeis e proporciona algunhas funcións para a súa manipulación:

```
> restart;
> poli:=6*x*y^5+12*y^4+14*x^3*y^3-15*x^2*y^3+9*x^3*y^2-30*x*
  y^2-35*x^4*y+18*y*x^2+21*x^5;
```

$$poli := 14\,x^3\,y^3 + 6\,x\,y^5 + 21\,x^5 - 35\,x^4\,y + 9\,x^3\,y^2 - 15\,x^2\,y^3 + 12\,y^4 + 18\,x^2\,y - 30\,x\,y^2 \quad \textbf{(1.2.1)}$$

```
> sort(poli); #por defecto ordea por graos totais (suma de
  potencia de x + potencia de y) decrecentes
```

(1.2.2)

$$14x^3y^3+6xy^5+21x^5-35x^4y+9x^3y^2-15x^2y^3+12y^4+18x^2y-30xy^2 \quad (1.2.2)$$

```
> sort(poli, [x,y], plex,descending); # ordea os termos de
  forma alfabética (en inglés). Ve-la axuda do comando sort
  para ver tódalas posibilidades: plex=purely lexicographic
```

$$21x^5-35x^4y+14x^3y^3+9x^3y^2-15x^2y^3+18x^2y+6xy^5-30xy^2+12y^4 \quad (1.2.3)$$

```
> sort(poli, [x, y], plex, ascending);
    #` `ordea os monomios por potencias crecentes de x (con
    empates, por potencias crecentes de y)
```

$$12y^4-30xy^2+6xy^5+18x^2y-15x^2y^3+9x^3y^2+14x^3y^3-35x^4y+21x^5 \quad (1.2.4)$$

```
> sort(poli, [x, y], tdeg, descending);
    # tdeg=total degree: ordea os monomios por sumas de
    potencias de x e y decrecentes
```

$$14x^3y^3+6xy^5+21x^5-35x^4y+9x^3y^2-15x^2y^3+12y^4+18x^2y-30xy^2 \quad (1.2.5)$$

```
> sort(poli, [x, y], tdeg, ascending);
    # ordea os monomios por sumas de potencias de x e y
    crecentes
```

$$-30xy^2+18x^2y+12y^4-15x^2y^3+9x^3y^2-35x^4y+21x^5+6xy^5+14x^3y^3 \quad (1.2.6)$$

```
> sort(poli, [y, x], plex, descending);
```

$$6y^5x+12y^4+14y^3x^3-15y^3x^2+9y^2x^3-30y^2x-35yx^4+18yx^2+21x^5 \quad (1.2.7)$$

```
> sort(poli, [y, x], plex, ascending);
```

$$21x^5+18yx^2-35yx^4-30y^2x+9y^2x^3-15y^3x^2+14y^3x^3+12y^4+6y^5x \quad (1.2.8)$$

```
> sort(poli, [y, x], tdeg, descending);
```

$$6y^5x+14y^3x^3-15y^3x^2+9y^2x^3-35yx^4+21x^5+12y^4-30y^2x+18yx^2 \quad (1.2.9)$$

```
> sort(poli, [y, x], tdeg, ascending);
```

$$18yx^2-30y^2x+12y^4+21x^5-35yx^4+9y^2x^3-15y^3x^2+14y^3x^3+6y^5x \quad (1.2.10)$$

```
> collect(poli, x); # ordea o polinomio segundo as potencias
  de x
```

$$21x^5-35x^4y+(14y^3+9y^2)x^3+(18y-15y^3)x^2+(-30y^2+6y^5)x+12y^4 \quad (1.2.11)$$

```
> collect(poli, y);  # ídem de y
```

$$6\,x\,y^5+12\,y^4+(-15\,x^2+14\,x^3)\,y^3+(9\,x^3-30\,x)\,y^2+(-35\,x^4+18\,x^2)\,y+21\,x^5 \quad \textbf{(1.2.12)}$$

```
> coeff(poli, x^3); # coeficientes de x^3
```

$$14\,y^3+9\,y^2 \quad \textbf{(1.2.13)}$$

Funcións racionais

As funcións racionais exprésanse como cocientes de dous polinomios, sendo o denominador distinto de cero.

```
> restart;
> f:=x^2+3*x+2; g:=x^2+5*x+6; h:=f/g;
```

$$f:=x^2+3\,x+2$$

$$g:=x^2+5\,x+6$$

$$h:=\frac{x^2+3\,x+2}{x^2+5\,x+6} \quad \textbf{(1.3.1)}$$

```
> numer(%); # comando que devolve o numerador da expresión anterior
```

$$x^2+3\,x+2 \quad \textbf{(1.3.2)}$$

```
> denom(h); #comando de devolve o denominador da función racional h
```

$$x^2+5\,x+6 \quad \textbf{(1.3.3)}$$

Por defecto, MAPLE non simplifica as funcións racionais. As simplificacións só se realizan cando o programa recoñece factores comúns.

```
> ff:=(x-1)*f;
```

$$ff:=(x-1)\,(x^2+3\,x+2) \quad \textbf{(1.3.4)}$$

```
> gg:=(x-1)^2*g;
```

$$gg:=(x-1)^2\,(x^2+5\,x+6) \quad \textbf{(1.3.5)}$$

```
> ff/gg;
```

$$\frac{x^2+3\,x+2}{(x-1)\,(x^2+5\,x+6)} \quad \textbf{(1.3.6)}$$

Para forzar unha simplificación utilízase a función **normal**:

```
> normal(f/g);
```

$$\frac{x+1}{x+3} \quad \textbf{(1.3.7)}$$

```
> normal(ff/gg);
```

$$\frac{x+1}{(x+3)\,(x-1)} \quad \textbf{(1.3.8)}$$

Tamén se poden definir funcións racionais en varias variabeis.

```
> restart;
> f:=161*y^3+ 333*x*y^2+184*y^2+162*x^2*y+144*x*y+77*y+99*
```

```
x+88;
```

$$f := 161\,y^3 + 333\,x\,y^2 + 184\,y^2 + 162\,x^2\,y + 144\,x\,y + 77\,y + 99\,x + 88 \quad \textbf{(1.3.9)}$$

```
> g:=49*y^2+28*x^2*y+63*x*y+147*y+36*x^3+32*x^2+177*x+104;
```

$$g := 49\,y^2 + 28\,x^2\,y + 63\,x\,y + 147\,y + 36\,x^3 + 32\,x^2 + 177\,x + 104 \quad \textbf{(1.3.10)}$$

```
> f/g;
```

$$\frac{161\,y^3 + 333\,x\,y^2 + 184\,y^2 + 162\,x^2\,y + 144\,x\,y + 77\,y + 99\,x + 88}{49\,y^2 + 28\,x^2\,y + 63\,x\,y + 147\,y + 36\,x^3 + 32\,x^2 + 177\,x + 104} \quad \textbf{(1.3.11)}$$

```
> normal(%);
```

$$\frac{161\,y^3 + 333\,x\,y^2 + 184\,y^2 + 162\,x^2\,y + 144\,x\,y + 77\,y + 99\,x + 88}{49\,y^2 + 28\,x^2\,y + 63\,x\,y + 147\,y + 36\,x^3 + 32\,x^2 + 177\,x + 104} \quad \textbf{(1.3.12)}$$

Pódese realiza-la **descomposición en fraccións parciais** utilizando o comando **convert** e especificando a opción **'parfrac'**

```
> restart;
> f:=(x^5+1)/(x^4-x^2);
```

$$f := \frac{x^5 + 1}{x^4 - x^2} \quad \textbf{(1.3.13)}$$

```
> convert(f, 'parfrac', x);
```

$$x - \frac{1}{x^2} + \frac{1}{x - 1} \quad \textbf{(1.3.14)}$$